PharmaSUG 2014 - Paper HA10

Estimating Medication Adherence Using a Patient-Mix Adjustment Method

R. Scott Leslie, MedImpact Healthcare Systems, Inc., San Diego, CA

ABSTRACT

The Centers for Medicaid and Medicare Services (CMS) and several national health care quality organizations regard medication adherence as a major attribute of quality of care. The preferred method of measuring medication adherence is the Proportion of Days Covered (PDC) by medication(s) over a specified review period. Although PDC can be calculated fairly easily from pharmacy claims using relatively little data elements, patient's medication adherence is most likely confounded by patient demographics and other measurable and immeasurable factors. This paper explores the use of a patient-mix adjustment method to account for patient's characteristics and previous medication history. Included is a description of a macro used to calculate PDC and estimate medication adherence.

INTRODUCTION

This paper starts with a brief background describing the importance of adherence to chronic medications to frame the relevance of medication adherence research. Then a technical description of the most accepted method of measuring adherence, Proportion of Days Covered (PDC), is explained by visual diagram and code. Included in this section is a description of a simple macro that uses arrays to calculate proportion of days covered by medication over a specified time period. Then this paper discusses how patient characteristics may confound, or mask, associations between an exposure of interest (e.g., intervention) and outcome of interest (e.g., adherence). Finally, this paper discusses a potential method that accounts for patient-mix. This paper refers to my SGF 2007 paper entitled "Using Arrays to Calculate Medication Utilization" that describe the use of arrays to measure medication coverage.

MEDICATION ADHERENCE SIGNIFICANCE

The Medication and Compliance Special Interest Group of ISPOR, the International Society for Pharmacoeconomics and Outcomes Research defines adherence as "the extent to which a patient acts in accordance with the prescribed interval and dose of a dosing regimen" (Cramer, 2007). Put another way, medication adherence is how well people follow prescribed doses of medication.

Non-adherence to medication is a pandemic problem that varies by study population and therapeutic class. The World Health Organization estimates adherence to long-term therapy for chronic diseases at 50% in developed countries and categorizes potential reasons for non-adherence into 5 groups; patient, disease state, health system, therapy and socioeconomic (De Geest, 2003). The consequences of non-adherence include unnecessary morbidity and mortality, lost quality of life, and costs estimated at $177 billion per year (Ernst, 2007).

CMS developed their Plan Star Ratings Program to measure health plan performance and allow beneficiaries to compare cost and quality of available Medicare Advantage Prescription Drug (MAPD) plans and Prescription Drug Plans (PDPs). CMS recognized poor adherence as a major public health problem and placed more importance on measures related to adherence. Of the 18 total Part D domain measures, the three Patient Safety Measures; Medication Adherence for Oral Diabetes Medications (D16), Medication Adherence for Hypertension (D17), and Medication Adherence for Cholesterol (D18) carry triple the weight of other measures and therefore contribute 11.7% of a plan's overall Star Rating and 31% of its Part D rating.

ADHERENCE ESTIMATED BY PROPORTION OF DAYS COVERED

The most common measurement method of medication adherence is Proportion of Days Covered, PDC. Basically this observational, indirect method uses pharmacy claims to calculate the days a person is covered by medication. There are limitations with this measurement method, namely, observed adherence may differ from actual adherence, but this is the preferred method endorsed by national organizations, such as CMS, NCQA, and Pharmacy Quality Alliance (PQA).

Below is a graphical representation of a patient's prescription claim history. Medication coverage in a measurement period can be estimated by using date of fill and days' supply data fields. The measurement period can be cross-sectional, which is the method used by CMS in their Part D Star Ratings adherence measures or longitudinal by following patients for similar lengths of time. This example shows 11 prescription claims for the patient, with claim 7

being shifted forward by the number of overlapping days of Claim 6 and claim 11 filled during measurement period with days' supply carrying outside the measurement period. PDC is the proportion of days covered to days reviewed. The first fill date is the beginning of a patient's review period. The date of disenrollment or end of measurement period is end of a patient's review period. In this example days of review is from 1/11/2011 to 12/31/2011, and days of coverage is 320 days. Therefore 320 days covered / 355 days reviewed yields a PDC of 90.1%, considered good adherence based on the most commonly used threshold (≥ 80%).



Display 1. Estimating Adherence via Proportion of Days Covered (PDC)

My paper entitled "Using Arrays to Calculate Medication Utilization", presented at 2007 SAS Global Forum, shows how arrays can calculate PDC by identifying medication coverage for all days in a review period. An extension of this code is described below and that included in the Appendix. In this example the study design is cross-sectional and patient adherence is estimated for a calendar year.

STEP 1

The first step is to prepare a prescription claims data set that contains (at the least) patient identifier, medication name, date of fill and days' supply variables. A quick SQL procedure calculates the maximum number of claims in the population used to assign a macro variable for use in a later data step. Here I create the macro variables outside of the macro for demonstration purposes.

```
proc sql;
create table cntclms as
       select mbr_id, count(*) as cnt_clms
              from claims
                     group by mbr_id
                            order by cnt_clms descending;
quit;

/*create macro variable for max claim count*/
data _null_;
set cntclms (obs=1);
call symput('clmsize', cnt_clms);
run;

%put "maximum claim size = " &clmsize.;
```

```
%let drug=cv_lipo;
%let nrclms=&clmsize.
%let id_start_dt='01-JAN-2011';
%let id_end_dt='31-DEC-2011'
%let postdays=365;
```

STEP 2

The second step is to transpose the prescription claims data from multiple observations per patient to a single observation per patient data set. The TRANSPOSE procedure is used twice to create fill dates and corresponding days' supply variables for each prescription claim. It is essential to sort the data set by patient and fill date.

```
proc transpose data =&indsn.(where=(drugname = "&drug."))
       out= fill_dt_&drug. (drop=_name_ ) prefix = fill_dt;
       by mbr_id drugname;
       var fill_dt;
run;

proc transpose data = &indsn.(where=(drugname = "&drug."))
       out=dsup_&drug. (drop=_name_ ) prefix = dsup;
       by mbr_id;
       var days_supply;
run;
```

STEP 3

Next, a data step uses arrays and DO loops to find the days of medication coverage for each patient in the review period. The first array, day, creates a dummy variable for each day in the review period. The next two arrays, groups the fill_dt and days_supply variables setting up the DO loops. The first DO loop sets each dummy variable, day1-day365, to 0. The second DO loop shift claims when claims (day of supply) overlap. The third DO loop uses an IF statement to flag the days of the review period that the patient was supplied the medication. The resulting data set contains 365 dummy variables, one for each day of the time period, indicating medication coverage.

```
data pdc_&drug.(keep=mbr_id fill_dt1 drugname day1 - day&postdays. );
       merge fill_dt_&drug.(in=a) dsup_&drug.(in=b);
       by mbr_id;

       if a=1 and b=1;
       array fill_dt(*) fill_dt1 - fill_dt&nrclms.;
       array dsup(*) dsup1 - dsup&nrclms.;
       array day(&postUsedays.) day1 - day&postdays.;

       do i=1 to &postdays.;
              day(i)=0;
       end;

       do i=2 to &nrclms. while (fill_dt(i) ne .);
              if fill_dt(i) < fill_dt(i-1) + dsup(i-1) then
                     fill_dt(i)= fill_dt(i-1)+ dsup(i-1);
       end;

       do ii=1 to &postdays.;
              do i = 1 to dim(fill_dt) while (fill_dt(i) ne .);
              if fill_dt(i)  <= &id_start_dt.d + ii -1 <= fill_dt(i) + dsup(i) -1 then
              day(ii)=1;
              end;
       end;

       drop i ii;
run;
```

STEP 4

This last data step calculates the numerator, day covered, and denominator, days of review for each patient which is then used to calculate PDC for each patient**.** PDC is calculated as days of medication coverage divided by day of

review. An indicator variable is also calculated using the 80% threshold commonly used to indicate adherence. The macro for this code is in the Appendix.

```
data pdc_&drug._2;
format id_end_dt date9. pdc_&drug. 9.6 pdc_pct_&drug. 10.6;
set pdc_&drug.;
by mbr_id;
id_end_dt=&id_end_dt.d;
review_days_&drug. = id_end_dt - fill_dt1 + 1;
days_covered_&drug. = sum(of day1 - day&postdays.);

if review_days_&drug. > 0 then
        do;
                pdc_&drug.      = days_covered_&drug. / review_days_&drug.;
                pdc_pct_&drug.  = round(pdc_&drug. * 100, .2);
                flag_pdc_&drug. = (pdc_&drug. >= 0.8);
        end;
run;
```

FACTORS RELATED TO ADHERENCE: ADJUSTING FOR CONFOUNDING

Medication adherence as estimated by PDC can be studied as the outcome of interest or the exposure of interest in an analysis. In most of my research, adherence is the outcome of interest, while an intervention or a change in the pharmacy benefit, is the exposure of interest. The intervention could be patient counseling or a communication to the patient or provider. To better estimate the effects of the intervention on adherence it is necessary to control for factors that may influence adherence. These potential confounders could be patient characteristics that are associated with both the exposure and outcome (as shown in Figure 1) and therefore may alter the association between the intervention and adherence. Confounding can be addressed in the design or analysis phase of the evaluation. The most common approach is to adjust for these characteristics in the modeling stage of the analysis by inserting the factors as independent variables in the model.



Figure 1: Confounding Triangle

As mentioned in an earlier section, contributors to adherence include patient, disease state and health system (provider, benefit) characteristics. Many of these factors may or may not be collected in a data set and some are most likely immeasurable. Among identified risk factors, comorbidity, or the presence and severity of disease, is a main factor that is often used for risk adjustment in payments. Accounting for comorbidity and other patient characteristics could be used to adjust for patient mix when estimating adherence.

ESTIMATING COMORBIDITY

There are several available measures of disease burden that are often used for risk adjustment. Many are indexes or scores that sum comorbidity indicators into a single score and subsequently used as a variable in modeling. Methods to estimate comorbidity vary in complexity. Complex measures use medical or pharmacy claims to estimate disease status for individuals. The Charlson Comorbidity Index (CCI) creates indicators for individual comorbidities based on ICD-9 codes to create a weighted overall score that ranges between 0 and 41 (higher scores indicating higher disease burden). The CCI was first developed to predict 10 year mortality risk but adaptations (e.g., Romano and Deyo) have been developed for various circumstances. Another instrument that uses ICD-9 codes was created

by Iezzoni *et al.* to identify severe forms of 13 chronic conditions common among the elderly. The AHRQ Clinical Classification System (CCS) which classifies ICD-9-CM diagnosis codes into 259 clinically meaningful and mutually exclusive categories has also been used as a risk adjuster. The pharmacy-based risk adjusters include MedicaidRx and RxRisk, both of which use pharmacy claims to estimate the patient's disease state. Alternatively, simple measures such as counting the number of drugs or distinct number of ICD-9 codes have been used. Research shows that these simple methods perform similar to other measures in certain populations and outcomes.

CASE STUDY: ADJUSTING FOR PATIENT-MIX

Applying the concepts above, I conducted an exploratory analysis to assess the extent to which patient characteristics, comorbidity and previous medication use affect adherence to antihypertensive medications. The objectives of this analysis were 1) to assess effect of patient-mix on adherence and 2) develop a potential risk adjustment tool. This tool could calculate an adherence value for a patient accounting for disease severity and other patient characteristics. The potential uses of this tool include reporting and predicting medication adherence at the member level and population level adjusting for potential confounders, identifying patients for targeted interventions and comparison of adherence rates across populations.

This observational analysis used administrative pharmacy claims data from a Medicare population using antihypertensive medications during the 2012 calendar year. The primary outcome was adherence as defined as PDC ≥ 80%. Figure 2 lists some of the factors available in the data set. Findings of published interventions have shown some of these factors can be both hazardous and protective.

Factor	Risk	Protective
Age	◊	◊
Gender	◊	◊
Comorbidity	◊	
Drug count (high)		◊
Low income status	◊	
Number of providers (high)	◊	
Filled 90-day supply		◊
Primary language ≠ English	◊	
New to therapy	◊	
History of good adherence		◊

Figure 2: Potential Confounders

Since the outcome is dichotomous, logistic regression was used to predict adherence to therapy over the 1-year measurement period. The first model included age as the only independent variable. Subsequent models incorporated the stepwise addition of other potential confounders in order to show the contribution of each factor. The final model included all factors, regardless of statistical significance of each, to assess extent of predicting adherence for patients. The models were evaluated on the ability to predict adherence, or the ability of each model to correctly distinguish adherent patients from non-adherent patients. Models were assessed using the *c*-statistic, which equals the area under the receiver operating characteristic (ROC) curve. *C*-statistics range from 0.5 – 1.0 with higher values indicating better ability of the model to discriminate between patients with and without the outcome. Values less than 0.7 are considered poor and those over 0.8 are considered good.

Here is the code for the final model. PROC LOGISTIC was used to model the probability of patients being adherent.

```
ods graphics on;

proc logistic data=model plots(only MAXPOINTS=NONE)=roc;
        class gender_cd lis_flag (ref='0') prim_lang_cd (ref='EN') flag_fill90 (ref='0')
        flag_newstart (ref='0') flag_pdc_base (ref='0');
        model flag_pdc (event='1') = age gender_cd lis_flag rxrisk_sum prscbr_cnt
        prim_lang_cd flag_fill90 flag_newstart flag_pdc_base

                / scale=none
                clparm=wald
                clodds=pl
                rsquare;
```

```
        units age=5 rxrisk_sum =2 prscbr_cnt=1;
run;
ods graphics off;
```

Figure 3 show results of entire model with all factors retained regardless of statistical significance. Baseline adherence was the largest predictor (OR = 4.95, 95% CI [4.31-5.96]) and the area under the ROC curve increased the most, from 0.66 to 0.74, when added to the model (Figure 4). Although this is a decent *c*-statistic, baseline adherence may not be present in all data sets, so additional factors related to adherence could add value to the model.

This analysis may be limited because if was performed in one population and the data most likely does not capture all information to account for the many potential confounders. Additional factors could include patient attitudes, provider relationships, pharmacy benefits, etc. Future research should include similar exploratory analyses using medical claims data and other factors to ascertain additional patient characteristics or factors that may be related to adherence. Also, applying this approach to other data sets and comparison of actual to predicted adherence values would provide more ways to create a risk-adjusted PDC.

Odds Ratio Estimates and Profile-Likelihood Confidence Intervals				
Effect	**Unit**	**Estimate**	**95% Confidence Limits**	
Age	5.0000	1.036	0.996	1.078
Gender F vs. M	1.0000	0.906	0.791	1.037
Low-Income Subsidy Member, yes vs. no	1.0000	1.005	0.871	1.160
Drug Count	2.0000	1.099	1.010	1.197
Number of Providers	1.0000	0.953	0.921	0.985
Primary Language Code, Spanish vs. English	1.0000	0.794	0.669	0.940
Primary Language Code, Other vs. English	1.0000	1.042	0.863	1.258
Filled 90-day supply, yes vs. no	1.0000	2.301	2.006	2.643
New to therapy, yes vs. no	1.0000	0.897	0.717	1.124
Adherence at Baseline, yes vs. no	1.0000	4.951	4.311	5.693

Figure 3: Odds Ratio Estimates and Profile-Likelihood Confidence Intervals of Final Model



Figure 4: ROC Curve of Final Model

CONCLUSION

Estimating medication adherence via PDC is fairly simple but is influenced by multiple patient, provider, environmental factors. Addressing these factors can reduce potential confounding when assessing associations between adherence and exposures or outcomes. One approach to limiting confounding is to create a risk adjuster based on patient characteristics, medication history and other available study variables.

REFERENCES

Cramer JA, Roy A, Burrell MBA, Fairchild CJ, Fuldeore MJ, Ollendorf, DA. Medication compliance and persistence: Terminology and definitons. Value in Health 2007; 11(1):44-47.

De Geest S, Sabaté E. Adherence to long-term therapies: evidence for action. World Health Organization, 2003;2(3):323. Available at: http://www.who.int/chp/knowledge/publications/adherence_full_report.pdf.

Ernst FR, Grizzle AJ. Drug-related morbidity and mortality: updating the cost-of-illness model. Journal of the American Pharmaceutical Association WashingtonDC 1996. 2001;41(2):192–199. Available at: http://www.ncbi.nlm.nih.gov/pubmed/11297331.

International Society of Pharmacoeconomics and Outcomes Research (ISPOR) Medication Compliance and Persistence Special Interest Group. http://www.ispor.org/sigs/medication.asp

Leslie, RS. Using arrays to calculate medication utilization. Proceedings of the 2007 SAS Global Forum, Orlando, FL.Paper 043-2007. Available at: http://www2.sas.com/proceedings/forum2007/043-2007.pdf

Radley DC, Gottlieb DJ, et al. Comorbidity risk-adjustment strategies are comparable among persons with hip-fracture. J Clin Epidemiol. 2008 61(6):580-587.

RECOMMENDED READING

- Leslie, RS. Using arrays to calculate medication utilization. Proceedings of the 2007 SAS Global Forum, Orlando, FL. Paper 043-2007. Available at: http://www2.sas.com/proceedings/forum2007/043-2007.pdf

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Name: R. Scott Leslie
Enterprise: MedImpact Healthcare Systems, Inc.
Address: 10680 Scripps Gateway Court
City, State ZIP: San Diego, CA 92131
Work Phone: 858-790-6685
E-mail: scott.leslie@medimpact.com
E-mail2: scottleslie1@yahoo.com

Visit my user page on SaaCommunity.org, http://www.sascommunity.org/wiki/User:Rsleslie

APPENDIX

```
/*Prepare for covdays macro*****************************************/

/*get max # of claims and apply it to define the array size ****/
proc sql;
        create table cntclms as
                select mbr_id, count(*) as cnt_clms
                        from claims
                                group by mbr_id
                                        order by cnt_clms descending;
quit;

/*create macro variable for max claim count*/
data _null_;
        set cntclms (obs=1);
        call symput('clmsize', cnt_clms);
run;

%put "maximum claim size = " &clmsize.;


/*PDC macro*********************************************************/
%macro covdays(indsn=, drug=, nrclms=, id_start_dt=, id_end_dt=, postdays=);

        proc transpose data =&indsn.(where=(drugname = "&drug."))
                out= fill_dt_&drug. (drop=_name_ ) prefix = fill_dt;
                by mbr_id drugname;
                var fill_dt;
        run;

        proc transpose data = &indsn.(where=(drugname = "&drug."))
                out=dsup_&drug. (drop=_name_ ) prefix = dsup;
                by mbr_id;
                var days_supply;
        run;

        data pdc_&drug.(keep=mbr_id fill_dt1 drugname day1 - day&postdays. );
                merge fill_dt_&drug.(in=a) dsup_&drug.(in=b);
                by mbr_id;

                if a=1 and b=1;
                array fill_dt(*) fill_dt1 - fill_dt&nrclms.;
                array dsup(*) dsup1 - dsup&nrclms.;
```

```
                array day(&postdays.) day1 - day&postdays.;

                do i=1 to &postdays.;
                        day(i)=0;
                end;

                do i=2 to &nrclms. while (fill_dt(i) ne .);
                        if fill_dt(i) < fill_dt(i-1) + dsup(i-1) then
                                fill_dt(i)= fill_dt(i-1)+ dsup(i-1);
                end;

                do ii=1 to &postdays.;
                        do i = 1 to dim(fill_dt) while (fill_dt(i) ne .);
                                if fill_dt(i)  <= &id_start_dt.d + ii -1 <= fill_dt(i) + dsup(i) -1
                                then day(ii)=1;
                        end;
                end;

                drop i ii;
        run;

        data pdc &drug. 2;
                format id_end_dt date9. pdc_&drug. 9.6 pdc_pct_&drug. 10.6;
                set pdc_&drug.;
                by mbr_id;
                id_end_dt=&id_end_dt.d;
                review_days_&drug. = id_end_dt - fill_dt1 + 1;
                days_covered_&drug. = sum(of day1 - day&postdays.);

                if review_days_&drug. > 0 then
                        do;
                                pdc_&drug.      = days_covered_&drug. / review_days_&drug.;
                                pdc_pct_&drug.  = round(pdc_&drug. * 100, .2);
                                flag_pdc_&drug. = (pdc_&drug. >= 0.8);
                        end;
        run;

%mend covdays;

/*Call macro*****************************************************************/
%covdays(indsn=claims, drug=cv_lipo, nrclms=&clmsize., id_start_dt='01-JAN-2011', id_end_dt='31-
DEC-2011' ,postdays=365);
```